

Un amplificateur de microphone Sparkfun

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :

[page d'accueil](#)

d'autres pdf, sur différents sujets :


[liste des PDF](#)

La société Sparkfun commercialise un circuit imprimé équipé d'un microphone électret et d'un amplificateur à base de l'OPA344. Ce montage, analogique, permet de réaliser un système audio faible coût. Nous allons vérifier, par la simulation, ses caractéristiques.


1) Présentation du circuit imprimé Sparkfun

<https://www.sparkfun.com/products/9964>

Le circuit est de petite taille avec le microphone soudé sur une face, et les composants CMS sur l'autre face. 3 traversées pour souder des fils sont installées : 2 pour l'alimentation (0 V ; VCC, de 2,7 V à 5,5 V) et 1 pour la sortie analogique (AUD). Le schéma est fourni :



Le circuit imprimé



a) Analyse du schéma de l'amplificateur audio

Il s'agit d'un montage à base d'un seul amplificateur opérationnel, l'OPA344 de TI (origine Burr Brown).

Ce dernier est monté en une structure classique d'amplificateur inverseur ($-R_5/R_4 = -100$).

La présence de C_3 entraîne une coupure haute.

L'alimentation étant monopolaire, un potentiel $VCC/2$ est placé en $e+$ par le pont diviseur R_3, R_2 . C_2 assure un découplage.

La « capsule microphone » nécessite une résistance de pull up pour fonctionner : c'est le rôle de R_1 . C_1 coupe la composante continue, ce qui est forme une coupure basse.

Cela totalise 10 composants + microphone.


b) Le microphone utilisé

Il est du type électret. Ses caractéristiques sont indiquées ci-dessous :

Omni-Directional Foil Electret Condenser Microphone							
DESCRIPTION							
Omni-Directional Foil Electret Microphone, 9.7 mm diameter and 4.5 mm high, Power Supply 5.0 V max, External Resistance Loading of 680 Ω, and sensitivity of -44 dB. Terminated with 2 solder points, Lead Free RoHS Compliant							
SPECIFICATIONS:							
Direction	Omni Directional Foil Electret			Minimum Direction sensitivity			
Operating Voltage Range	Vs= 1.0 Vdc ~ 10.0 Vdc			Power Supply (Vs)		1.5 V	
Frequency Range	100 ~ 10,000 Hz.			Maximum Current		0.5 mA	
Sensitivity	- 46 ± 2.0, (0 dB = 1V / Pa) at 1K Hz.			Minimum Sensitivity to Noise Ratio		58 dB	
Sensitivity Reduction	3.0 V to 2.0 V -3 dB			Maximum input S.P.L.		110 dB at 1.0 KHz, THD <1%	
Operating Temperature	-20°C to + 60°C			Storage Temperature		-40°C to + 75°C	
Loading Resistance (RL)	External, 680 Ω at Vs = 1.5 V, Max. 2,200 Ω			Built in Capacitors		None	
Termination	PC Pins, 4.5 mm Long, 0.6 mm ø, 2.54 mm Spacing						
Dimensions	Length / Diameter	9.7 mm ø	Height	4.5 mm	Housing Material	Al-Mg Alloy. Color	
Approximate Weight	0.7 grams	Options				Compliance	RoHS, Lead Free

On peut lire que sa bande passante est environ [100 Hz ; 10 kHz], ce qui couvre largement le spectre de la voix humaine. Avec une alimentation de 1,5 V, il nécessite une résistance de pull up $R_L = 680 \Omega$. Cette tension peut aller de 1 V et 10 V. Il est préconisé la valeur maximale à R_L : 2,2 kΩ.


En fait, la capsule est constituée en interne de 2 étages : Le capteur sonore d'une part, et un préamplificateur associé d'autre part. Ce pré-ampli est à base de transistor à effet de champ, ce qui permet de présenter une faible impédance de sortie.



C'est à l'utilisateur de connecter le pull up et le condensateur de liaison, ce qui est fait dans le circuit imprimé de Sparkfun par R_1 , C_1 respectivement.

c) L'amplificateur opérationnel OPA344

Il est de la famille "Low Power". En boîtier SOT23-5 : 3 bornes (e+, e-, sortie) et 2 pour l'alimentation. Il n'a donc pas de réglage d'offset, ni de compensation en fréquence.



Il est donné pour 120 à 122 dB typique de gain statique en boucle ouverte, valeur relativement élevée.

PARAMETER	CONDITION	MIN	TYP	MAX	UNITS
OPEN-LOOP GAIN Open-Loop Voltage Gain A_{OL}	$R_L = 100k\Omega, 10mV < V_O < (V+) - 10mV$	104	122		dB
	$R_L = 5k\Omega, 400mV < V_O < (V+) - 400mV$	96	120		dB

Un autre extrait de la datasheet montre une fréquence de transition de 1 MHz, un slew rate de 0,8 V/μs, ce qui est très modeste pour un A. Op, mais suffisant pour notre application.

Le point positif : l'étage de sortie (amplificateur à transistor CMOS en classe AB) permet de travailler de 0 à VCC en rail to rail.

- RAIL-TO-RAIL INPUT
- RAIL-TO-RAIL OUTPUT (within 1mV)
- LOW QUIESCENT CURRENT: 150μA typ
- GAIN-BANDWIDTH
OPA344: 1MHz, $G \geq 1$
- SLEW RATE
OPA344: 0.8V/μs

Le modèle Pspice de cet A.Op. est disponible, dans le fichier OPA2344.lib.

2) Simulations

a) L'amplificateur opérationnel OPA344 testé séparément

Ces runs ont pour but de vérifier si le modèle Pspice de l'OPA344 est conforme aux spécifications du circuit réel.

Un **premier test** permet de vérifier la réponse harmonique de l'A. Op. seul, donnée en petits signaux.

Une solution, réalisable en simulation mais pas en pratique, consiste à tester l'A. Op. sans aucune contre réaction.

Il faut placer un potentiel continu sur les entrées pour tenir compte que l'alimentation est simple (et non symétrique).

Pour ne pas avoir la sortie en saturation (0 V ou +5 V), on ajoute une tension sur ϵ pour contrebalancer l'offset de l'OPA344.




Schéma simulé

DC = -187,48 μ V
(déterminé après plusieurs essais).

VS = 5 V
RL = 10 k Ω
CL = 100 pF

```
ampli op OPA344
* fichier test1_OPA344.cir
.lib OPA2344.lib ; = OPA344
.OPTION ITL4=100 VNTOL=1E-7
+ GMIN=1E-13 RELTOL=0.0001
Valim1 50 0 dc 5V
Valim2 25 0 dc 2.5V
Vmc emoins 0 DC 2.5
.param off= -187.48u
Veps eplus emoins DC {off} AC 1
RL S 25 10k
CL S 25 100p
X1 eplus emoins 50 0 S OPA2344
.OP
.AC dec 1000 0.1 2MEG
.PROBE
.END



Netlist, prête à simuler
```

Une première vérification est la polarisation correcte de l'A.Op, comme le montre cet extrait du fichier.out.

Rappelons qu'un décalage de 1 μ V en ϵ entraîne un décalage de 1 V en sortie ! D'où la précision à apporter pour la composante DC.

NODE	VOLTAGE
(S)	2.5008

La réponse harmonique peut donc être lancée :



Interprétation :

20 log du module : le gain statique simulé est de 116 dB, ce qui est conforme à la spécification, mais différent de 4 dB de la courbe donnée par le constructeur. Le passage à 0 dB est à 1 MHz, en simulation comme sur la datasheet.

Phase : la première cassure est un peu décalée : par exemple la traversée à -30° se situe à 1 Hz en simulation et vers 4 Hz sur la datasheet. A 2 MHz, la phase est à -161° en simulation et vers -150° donné par le constructeur.

En première conclusion, dans l'utilisation en régime harmonique, le modèle de l'OPA344, donne une concordance très raisonnable.

Cette caractéristique montre qu'en boucle fermée, avec 40 dB de gain, la bande passante sera de 10 kHz.

Un **deuxième test** permet de mesurer le slew rate.

Il s'agit d'une réponse en grands signaux.

La doc de TI n'indique pas le schéma utilisé pour ce test. Mais à partir des explications « texte » données, on peut déduire le schéma de test ci-contre.

Il s'agit du montage classique du suiveur.




Schéma simulé

VS = 5 V
 RL = 10 kΩ
 CL = 100 pF

ampli op OPA344
 * fichier test2_OPA344.cir

```
.lib OPA2344.lib ; = OPA344
.OPTION ITL4=100 VNTOL=1E-7
+ GMIN=1E-13 RELTOL=0.0001
Valim1 50 0 dc 5V
Valim2 25 0 dc 2.5V
Ve eplus 0 pulse (0.5 4.5 8u 10n 10n 27u 50u)
RL S 25 10k
CL S 25 100p
X1 eplus S 50 0 S OPA2344
.OP
.TRAN 10n 50u 0 10n
.PROBE
.END
```

Netlist, prête à simuler


Interprétation :

Le résultat de la simulation est présenté ci-contre superposé à la caractéristique donnée par le constructeur.

La très bonne coïncidence permet de vérifier que le macro modèle est conforme à la réalité.

On peut lire, en plaçant le curseur, 0,8 V/μs (montant et descendant), ce qui est conforme avec la documentation.

LARGE-SIGNAL STEP RESPONSE: OPA344
 G = +1, RL = 10kΩ, CL = 100pF



Les 2 courbes sont superposées, à la même échelle

Vérifions sur un **troisième run** l'aptitude à délivrer un signal allant de 0 à 5 V pour ce suiveur alimenté entre 0 V et 5 V. Il s'agit de vérifier le 5V « rail to rail ».

Cet essai peut se faire avec n'importe quel signal d'entrée, mais il est d'usage d'exploiter un signal sinusoïdal, porté par une composante continue. Par simplicité, on exploite l'A. Op. monté en suiveur.

En sortie, on attend un signal d'équation : $V_o + V_{max} \sin(2 \pi f t)$, avec $V_o = 2,5 V$ et $V_{max} = 2,5 V$

Mais il faut, pour ne pas être pénalisé par le slew rate, la condition : $f < \frac{SR}{2 \pi V_{max}} = \frac{0,8}{2 \pi 2,5} = 50,9 \text{ kHz}$

La datasheet montre un essai à 40 kHz, ce qui est conforme à cette condition.

On reprend le schéma précédent, et on change le signal d'entrée.





Schéma simulé

ampli op OPA344
 * fichier test3_OPA344.cir


```
(...)
Ve eplus 0 SIN (2.5 2.5 40k)
Une ligne à changer dans la netlist
```

La réponse transitoire peut donc être lancée :

Interprétation :




Pour faciliter la comparaison avec le chronogramme fourni par TI, on a présenté l'entrée d'une part, et la sortie inversée décalée d'autre part.




Le curseur montre la tension de sortie qui va de 27 mV à 4,97 V. On est donc, à 30 mV près environ, à 0 V ; 5 V.

b) Un modèle Pspice de microphone électret

La capsule est constituée d'un capteur formé tel un condensateur : 2 armatures séparées par un isolant. Une tension est entre ces armatures. Une des armatures est mobile, et se déplace en fonction de la pression liée à l'onde acoustique. Cela provoque une faible tension. Un transistor JFET monté en source commune permet de transmettre ce signal sous plus faible impédance. Cela donne les 2 fils de connexion.




Il faut donc une résistance de polarisation pour placer le JFET en mode source de courant contrôlée. C'est le rôle de R₁ dans le circuit imprimé.



Dans ce cas, le schéma équivalent en dynamique est, avec $i_D = g_m v_{GS}$:

Par application du théorème de Thevenin, le schéma équivalent de ce microphone électret associé à son amplificateur à JFET est une f.e.m. en série avec une résistance. Ici, la f.e.m. est fixée à 20 mV d'amplitude. La forme d'onde du signal est liée à l'onde acoustique, mais, pour un run de vérification, en place un signal sinusoïdal de fréquence 1 kHz. R₁ = 2,2 kΩ.




c) Simuler le circuit Sparkfun

Le macromodèle de l'OPA344 est très volumineux en composants, et le schéma de la carte Sparkfun dépasse tout juste le nombre max de composants pour la version d'évaluation de Pspice. En remplaçant le pont diviseur R₂ R₃ par une alim constante, le schéma devient simulable.

```

ampli + microphone mono tension
* fichier sparkfun.cir
.lib OPA2344.lib ; = OPA344
Valim 5 0 dc 5V ; pour ampli Op
Ve 6 0 sin (0 20m 1k) ac 1
R1 6 7 2.2k
C1 7 8 4.7u
R4 8 4 10k
VDC 3 0 DC 2.5
R5 4 1 1Meg
C3 4 1 12p
X1 3 4 5 0 1 OPA2344
.OP
.AC dec 100 0.1 1MEG
.TRAN 10u 2m 0 10u
.probe
.end
    
```

Netlist, prête à simuler




Le schéma simulé.

• **L'analyse .OP** permet de vérifier la polarisation. On lit sur le fichier de sortie $V(1) = 2,5002\text{ V}$. L'ampli op, par sa contre réaction statique, a équilibré e+ et e-. Il subsiste l'offset que nous avons déterminé à $-187,48\ \mu\text{V}$, que l'on peut faire glisser en série avec $2,5\text{ V}$, ce qui aboutit à la valeur théorique de $2,50018\dots\text{ V}$. On est en accord, à l'arrondi de l'affichage près.

• **L'analyse .AC** montre en sortie $V(1)$ (vis à vis de l'entrée qui est la f.e.m. placée en $V(6)$) : un gain de $38,2\text{ dB}$ sur le plateau, une coupure haute de 7 kHz , une coupure basse de $2,8\text{ Hz}$.

Measurement	Value
Cutoff_Lowpass_3dB(V(1))	7.05750k
Cutoff_Highpass_3dB(V(1))	2.78091

Une mesure à 1 kHz donne une amplification de module **81,174**.



On vérifie très bien que c'est C_1 qui forme la coupure basse.

Avec C_1 à $0,47\ \mu\text{F}$ (au lieu de $4,7\ \mu\text{F}$), le même run de simulation montre une coupure à $27,7\text{ Hz}$. Sparkfun aurait pu faire ce choix, car la capsule ne laisse passer les sons qu'à partir de 100 Hz . On peut dire qu'ici C_1 est surdimensionné.

En visualisant $\text{dB}(V(1)/V(8))$, on retrouve l'amplification de 40 dB , et une coupure à $6,39\text{ kHz}$. C'est la présence de C_3 qui provoque cette coupure.

Sans ce condensateur, la coupure du montage de gain 40 dB passe à $12,39\text{ kHz}$ (limité par la réponse de la boucle ouverte de l'OPA344), et la bande passante de la carte passe à $15,1\text{ kHz}$.

La capsule est donnée pour des signaux allant à 10 kHz , ce qui est dans la bande passante de l'OPA344 monté en amplification de 100 . Or Sparkfun ajoute C_3 pour brider sa bande à $6,39\text{ kHz}$, ce qui donne la fréquence max à -3 dB de la carte à 7 kHz .


Il est probable que ce choix de couper les hautes fréquences est pour retirer le bruit (du souffle dans les médium et aigus), et donc gagner en confort sur un écouteur.

• **L'analyse .TRAN** confirme les observations

A 1 kHz , l'analyse AC a montré que l'amplification de toute la chaîne était de **-81,174**.

On retrouve cette valeur : en entrée on applique 40 mV càc, et en sortie on mesure $3,2454\text{ càc}$ en opposition de phase, soit une **amplification de -81,135**. Le signal variable est porté par la composante continue de $2,5\text{ V}$.

Measurement	Value
Max(V(1))- Min(V(1))	3.24540



**entrée : 40 mV càc, 1 kHz .
sortie : $3,2454\text{ càc}$, porté par $2,5\text{ V}$**

Conclusion

Ces simulations ont exploité des notions classiques de l'électronique analogique. Ce document a permis d'exploiter un montage concret. Le modèle de l'OPA344 a permis de vérifier finement le fonctionnement, notamment la dynamique 5 V rail to rail.

articles 1 à 43 : sur le livre

Tableau récapitulatif des articles PDF disponibles sur ce site

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	présentation	document PDF
44	Exemples basiques et des exercices...	présentation	document PDF
45	Un exemple de circuit passif	présentation	document PDF
46	Un oscillateur Colpitts	présentation	document PDF
47	Compensation en fréquence des amplificateurs opérationnels	présentation	document PDF
48	Un amplificateur à transistors bipolaires	présentation	document PDF
49	Une bascule D Flip Flop CMOS	présentation	document PDF
50	Une porte XOR à transistors MOS	présentation	document PDF
51	Un VCO à 12 transistors MOS	présentation	document PDF
52	Une PLL à moins de 20 transistors MOS	présentation	document PDF
53	Un oscillateur à résistance négative	présentation	document PDF
54	Une charge électronique	présentation	document PDF
55	Un amplificateur en classe C	présentation	document PDF
56	Le monostable 74 123	présentation	document PDF
57	Un amplificateur en classe D	présentation	document PDF
58	Le transformateur en linéaire	présentation	document PDF
59	La loi d'ohm thermique	présentation	document PDF
60	Le transformateur en non linéaire	présentation	document PDF
61	Robustesse d'un oscillateur en anneau	présentation	document PDF
62	Une alimentation stabilisée	présentation	document PDF
63	Modélisation d'un haut-parleur	présentation	document PDF
64	Un synthétiseur de fréquence	présentation	document PDF
65	Un ampli audio de Sparkfun	présentation	document PDF
66	Simulation logique et analogique	présentation	document PDF
67	Un oscillateur à relaxation	présentation	document PDF
68	Lecteur de TAG RFID 125 kHz	présentation	document PDF
69	Diagramme de l'œil avec Pspice	présentation	document PDF
70	Un amplificateur hautes fréquences	présentation	document PDF
71	Une bizarrerie enfin expliquée...	présentation	document PDF
72	Comprendre le paramétrage de la FFT	présentation	document PDF
73	La relation de Bennett	présentation	document PDF
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	présentation	document PDF
75	Une horloge biphase sans recouvrement	présentation	document PDF
76	Quelques simulations sur la diode	présentation	document PDF
77	Un ampli classe A, avec transformateur de sortie	présentation	document PDF
78	Des stimuli pour PSpice	présentation	document PDF
79	Simuler le TL431 : zener ajustable	présentation	document PDF
80	Un ADC flash	présentation	document PDF
81	Une chaîne d'acquisition : S&H, ADC, DAC	présentation	document PDF
82	Un amplificateur 50 MHz	présentation	document PDF
83	Un dérivateur non inverseur	présentation	document PDF
84	Un amplificateur bipolaire avec push pull CMOS	présentation	document PDF
85	Rôle des répéteurs logiques dans un circuit intégré	présentation	document PDF
86	Un driver logique CMOS pour charge 50 ohms	présentation	document PDF
87	Des triggers de Schmitt et des applications	présentation	document PDF
88	Un filtre gaussien analogique	présentation	document PDF
89	Un générateur de bruit rose	présentation	document PDF

90	Un anémomètre à fil chaud : simulation comportementale	présentation	document PDF
91	Un oscillateur à pont de Wien stabilisé par CTN	présentation	document PDF
92	L'emballement thermique d'une diode	présentation	document PDF
93	Les puissances dans un amplificateur	présentation	document PDF
94	Asservissement de puissance dans une résistance	présentation	document PDF
95	Asservissement de la puissance émise par une antenne radio	présentation	document PDF
96	Un driver de LED de puissance	présentation	document PDF
97	Exploiter Pspice pour simuler des filtres numériques	présentation	document PDF
98	Un filtre en cosinus surélevé avec Pspice	présentation	document PDF
99	Effet de la température sur un amplificateur en classe A	présentation	document PDF
100	Un amplificateur à transistors JFET et bipolaires	présentation	document PDF
Supplément, hors article :			
mon cours « Electronique pour les communications numériques », polycopié couleur 201 pages en pdf			

[retour à l'écran d'accueil de ce site](#)